Getting Started with Market Equilibrium Models

Thomas F. Rutherford

Department of Agricultural and Applied Economics
University of Wisconsin, Madison

WiNDC Short Course – 20 July 2021

Outline

- Paul Samuelson
- Activity Analysis and Market Equilibrium
- Application: Electricity Investment and Dispatch
- Spatial Price Equilibrium

Samuelson in 1950



Paul Samuelson

- University of Chicago (B.A.) – Harvard University (Ph.D.),
- Enrolled college at age 16
- Full professor at age 32
- First American to win the Nobel Memorial Prize in Economic Sciences: "[Samuelson] has done more than any other contemporary economist to raise the level of scientific analysis in economic theory."
- Recruited numerous Nobel laureates at MIT: Robert M. Solow, Paul Krugman, Franco Modigliani, Robert C. Merton and Joseph E. Stiglitz.

A Theory which is both True and Nontrivial

Stanislaw Ulam once challenged Samuelson to name one theory in all of the social sciences which is both true and nontrivial.

A Theory which is both True and Nontrivial

Stanislaw Ulam once challenged Samuelson to name one theory in all of the social sciences which is both true and nontrivial.

Several years later, Samuelson responded with David Ricardo's theory of comparative advantage:

> *That it is logically true need not be argued before a mathematician; that is not trivial is attested by the thousands of important and intelligent men who have never been able to grasp the doctrine for themselves or to believe it after it was explained to them.*

Outline

- Paul Samuelson
- Activity Analysis and Market Equilibrium
- Application: Electricity Investment and Dispatch
- Spatial Price Equilibrium

Activity Analysis

When there are a discrete set of production technologies, each characterized by a marginal cost and a capacity, the supply curve becomes a step function corresponding to the sorted sequence of plant capacities.

Consider a market in which the commodity is supply by the following four technologies:

	c_j	k_j
a	2	2
b	5	2
c	7	4
d	10	∞

Activity Analysis Supply Curve



Market Equilibrium with Activity Analysis

Consider a market equilibrium when there are multiple discrete supply technologies. As in the conventional continuous Marshallian model, the equilibrium price and quantity is defined by the intersection of the supply and demand schedules:



Market Equilibrium and Social Surplus

A convenient property of the competitive market allocation is that it *maximizes* social surplus, as illustrated in this figure:



Constrained Optimization Approach

Let $Q_t \geq 0$ denote output from technology t, P denote the equilibrium price, PS and CS denote producer and consumer surplus. The market equilibrium then solves:

$$\max PS + CS$$

subject to:

- Market supply equals technology output:

$$S = \sum_t Q_t$$

- Market equilibrum price is on the demand curve:

$$P = 10 - \frac{5}{6} S$$

- Producer surplus is the area below the market price and above the cost of production: $PS = \sum_t (P - c_t) Q_t$
- Consumer surplus is the area under the demand curve: $CS = \frac{(10-P)}{2} S$

Geometric Interpretation of the Equilibrium



GAMS Code – Sets and Data

```
$title  surplus maximization and market equilibrium

set     t /a,b,c,d/;

table   tech    Technology
        cost    cap
a       2       2
b       5       2
c       7       4
d       10      inf;

parameter       c(t)    Cost by technology;
c(t) = tech(t,"cost");
```

GAMS Code – Variable Declaration

```
nonnegative variables   P,PS,CS,s,Q(t);
free variable           obj;
equations               price, supply, psurplus, csurplus, objective

price..         P =e= 10 - S*10/6;

supply..        S =e= sum(t, Q(t));

psurplus..      PS =e= sum(t, (P-c(t))* Q(t));

csurplus..      CS =e= (10 - P)*S/2;

objective..     OBJ =e= CS + PS;

Q.UP(t) = tech(t,"cap");

model equil /all/;
solve equil using nlp maximizing OBJ;
```

GAMS Listing File

```
                    LOWER          LEVEL          UPPER         MARGINAL
---- VAR P           .             5.0000       +INF              .
---- VAR PS          .             6.0000       +INF              .
---- VAR CS          .             7.5000       +INF              .

---- VAR Q
a                    .             2.0000        2.0000         3.0000
b                    .             1.0000        2.0000         EPS
c                    .              .            4.0000        -2.0000
d                    .              .           +INF           -5.0000

---- VAR obj        -INF            13.5000     +INF              .
```

Outline

- Paul Samuelson
- Activity Analysis and Market Equilibrium
- Application: Electricity Investment and Dispatch
- Spatial Price Equilibrium

A Load-Duration Curve

A load-duration curve portrays electricity demand over a year in terms of sorted decreasing quantity. Typically constructed on an hourly basis (8760 hours per year):



Load Segment Approximation of a Load-Duration Curve

If we want to model electricity sector investment decisions, we need to work with an approximation to the load-duration curve. It does not take too many *load segments* to produce a coherent representation:



Units Characteristics Depend on Load Factors

- *Peak* generating units typically operate a small number of hours per year and tend to have low capital costs and high variable costs.
- *Base load* generating units typically operate a large number of hours per year and tend to have high capital costs and low variable costs.



Load Dispatch Curve

Investment and dispatch decisions are made jointly: when a utility invests in new generating capacity, it must take into account overall load duration curve and characteristics of existing capacity:



A Canonical Electricity Investment Model

- Sets

  s Load segments as illustrated above.

  j Generating units, e.g. existing capacity, new investment options

  i Demand categories, e.g. residential, commercial, industrial

  f Fuel types, e.g. hard coal, soft coal, natural gas, uranium

Market Data

h_s Segment durations, hours

$\bar{p}_s$, $\bar{D}_{is}$, ϵ_{is} Demand characteristics as might be represented by representative price-quantity paris and elasticities of demand (price expressed in € per KW, demand in KW and elasticity is dimensionless)

Unit Level Data

- ϕ_{fj} Heat rates describing input fuel requirements per unit generation (PJ per KWH)
- $\bar{K}_j$ Capacities of existing generating units, TW
- c_f Fuel costs (€ per PJ)
- α_{js} Average availability factor for generating units, reflecting need for repair and intermittency of renewable energy sources (dimensionless)
- r_j^K Rental price of *new* generating capacity, (€ per KW per year), typically computed on the basis of capital cost, depreciation rate, capital cost and fixed maintenance and operating costs:

$$r_j^K = \begin{cases} p_j^K(r+\delta) + c_j^M & \text{New plants} \\ c_j^M & \text{Extant plants} \end{cases}$$

- p_j^M Variable maintenance and operating costs, (€ per KWH)

Variables which Determine an Equilibrium

- Primal Variables : quantities

  X_{js} Generation and dispatch
  K_j Generating utilization (extant and new vintage)

- Dual Variables : prices

  p_s Wholesale prices by load segment
  π_{js} Profit margins
  μ_j Shadow price on installed (extant) capacity

Primal Equilibrium Conditions

- Aggregate demand:

$$D_s = \sum_i \bar{D}_{is} \left(1 - |\epsilon_{is}|(p_s/\bar{p}_s - 1)\right)$$

Primal Equilibrium Conditions

- Aggregate demand:

$$D_s = \sum_i \bar{D}_{is} \left(1 - |\epsilon_{is}|(p_s/\bar{p}_s - 1)\right)$$

- Market clearance:

$$D_s = \sum_j X_{js} \quad \perp \quad p_s$$

Primal Equilibrium Conditions

- Aggregate demand:

$$D_s = \sum_i \bar{D}_{is} \left(1 - |\epsilon_{is}|(p_s/\bar{p}_s - 1)\right)$$

- Market clearance:

$$D_s = \sum_j X_{js} \quad \perp \quad p_s$$

- Feasibility of generation:

$$\alpha_{js} K_j \geq X_{js} \geq 0 \quad \perp \quad \pi_{js} \geq 0$$

Primal Equilibrium Conditions

- Aggregate demand:

$$D_s = \sum_i \bar{D}_{is} \left(1 - |\epsilon_{is}|(p_s/\bar{p}_s - 1)\right)$$

- Market clearance:

$$D_s = \sum_j X_{js} \quad \perp \quad p_s$$

- Feasibility of generation:

$$\alpha_{js} K_j \geq X_{js} \geq 0 \quad \perp \quad \pi_{js} \geq 0$$

- Capacity:

$$\bar{K}_j \geq K_j \quad \perp \quad \mu_j \geq 0$$

Dual Equilibrium Conditions

- Profitability – arbitrage in dispatch:

$$\pi_{js} + p_j^M + \sum_f c_f \phi_{fj} \geq p_s \quad \perp \quad X_{js} \geq 0$$

Dual Equilibrium Conditions

- Profitability – arbitrage in dispatch:

$$\pi_{js} + p_j^M + \sum_f c_f \phi_{fj} \geq p_s \quad \perp \quad X_{js} \geq 0$$

- Profitability – arbitrage in investment:

$$r_j^K + \mu_j \geq \sum_s h_s \alpha_{js} \pi_{js} \quad \perp \quad K_j \geq 0$$

Integrability: Equilibrium Allocation = Optimal Allocation

$$\begin{aligned} \max \quad & \textstyle\sum_{i,s} \bar{p}_s D_{is} \left(1 + 1/|\epsilon_{is}|(1 - \frac{D_{is}}{2\bar{D}_{is}})\right) \\ & - \textstyle\sum_{sj} X_{js} h_s \left(\sum_f c_f \phi_{jf} + p_j^M\right) - \sum_j K_j r_j^K \end{aligned}$$

subject to:

$$\sum_{is} D_{is} = \sum_{j} X_{js}$$

$$\alpha_{js} K_j \geq X_{js} \geq 0$$

$$\bar{K}_j \geq K_j$$

$$K_j \geq 0$$

Outline

- Paul Samuelson
- Activity Analysis and Market Equilibrium
- Application: Electricity Investment and Dispatch
- Spatial Price Equilibrium

Spatial Price Equilibrium

i	Supply nodes
j	Demand nodes
c_{ij}	Unit shipment costs
μ_i	Unit (marginal) production cost
$\bar{S}_i$	Supply limit (upper bound)
$\bar{D}_j$	Demand quantity

Least Cost Production and Distribution

$$\min \sum_i \mu_i S_i + \sum_{i,j} c_{ij} X_{ij}$$

subject to:

$$\begin{aligned} S_i \geq & \quad \textstyle\sum_j X_{ij} \\ \sum_i X_{ij} \geq & \quad D_j \end{aligned}$$

$$D_j = \bar{D}_j, \quad S_i \leq \bar{S}_i$$

GAMS Code

```
$title  A Calibrated Spatial Price Equilibrium Model

$ontext

We first formulate a linear programming model which minimizes
the cost of production and distribution on a transportation
network with supply nodes and demand nodes.  Using the primal
and dual values from the LP model we calibrate an economic
equilibrium model with price elastic demand and supply for
which the reference equilibrium corresponds precisely to the
LP optimum.

$offtext

*       Generate a random instance of the problem:

set     i       Supply nodes /1*5/
        j       Demand nodes /1*5/;


parameter       d0(j)   Demands
                s0(i)   Supply
                mu(i)   Marginal cost of production,
                c(i,j)  Transport cost;

c(i,j) = uniform(0,1);
d0(j)  = round(uniform(1,100));
s0(i)  = round(uniform(1,200));
mu(i)  = uniform(0.5,1.5);
```

GAMS Code (cont)

```
*       Here I illustrate the lazy way to declare variables.  When
*       a variable is declared with no arguments, the dimensionality
*       is inferred at the first use and the domains are assumed
*       to be the universe, e.g. X(*,*).

*       The disadvantage of this approach is that domain errors
*       may be undetected and difficult to trace. It is a good idea
*       to use explicit domain whereever possible:

nonnegative variables   X,S,D;

free variable   TOTCOST         Objective function;

equations       objdef, supply, demand;

objdef..        TOTCOST =e= sum((i,j), c(i,j) * X(i,j)) + sum(i, mu(i)*S(i));

*       Orient both equations as >= so that the Lagrange multipliers
*       are non-negative:

supply(i)..     S(i) =g= sum(j, X(i,j));

demand(j)..     sum(i, X(i,j)) =g= D(j);

model transport /all/;

*       Fix demand and place an upper bound on supply in order
*       that the marginal cost of supply is included in the
*       shadow prices at the equilibrium point:

S.UP(i) = s0(i);  D.FX(j) = d0(j);

solve transport using LP MINIMIZING TOTCOST;
```

LP Solution

Formulated as a capacity-constrained supply with constant marginal cost, the shadow prices at supply and demand nodes reflect both the production and transportation costs:

```
---- EQU supply

         LOWER          LEVEL          UPPER         MARGINAL

1          .              .            +INF            1.3821
2          .              .            +INF            1.3298
3          .              .            +INF            1.2227
4          .              .            +INF            1.1282
5          .              .            +INF            1.2468

---- EQU demand

         LOWER          LEVEL          UPPER         MARGINAL

1          .              .            +INF            1.5539
2          .              .            +INF            1.2878
3          .              .            +INF            1.3783
4          .              .            +INF            1.3969
5          .              .            +INF            1.3534
```

Calibrated Supply and Demand Functions



Calibrated Supply and Demand Functions

Given the following *data*:

- $\bar{q}$ Reference quantity supplied (and demanded)
- $\bar{p}$ Reference demand price
- $\bar{\mu}$ Reference supply price
- ϵ Magnitude of the price elasticity of demand
- η Magnitude of the price elasticity of supply

We can write the demand and supply functions as:

$$d(p) = \bar{d}\left(1 - \epsilon\left(\frac{p}{\bar{p}} - 1\right)\right)$$

and

$$s(\mu) = \bar{s}\left(1 + \eta\left(\frac{\mu}{\bar{\mu}} - 1\right)\right)$$

Mathematical Facts

1. $$\frac{\mathrm{d}}{\mathrm{d}Q}\int_{q=0}^{Q} p(q)dq = p(Q)$$

2. The first order conditions for

$$\max \sum_i f_i(S_i) + \sum_j g_j(D_j)$$

s.t.

$$\begin{aligned} S_i \geq & \sum_j X_{ij} & \perp \mu_i \\ \sum_i X_{ij} \geq & D_j & \perp p_j \end{aligned}$$

are

$$\frac{\mathrm{d}f_i(S_i)}{\mathrm{d}S_i} = -\mu_i$$

and

$$\frac{\mathrm{d}g_j(D_i)}{\mathrm{d}D_i} = p_j$$

Integrable Demand

The calibrated inverse demand function corresponding to $D_j(p_j)$ is

$$p_j(q) = \bar{p}_j \left(1 + \left(1 - q/\bar{D}_j\right)/\epsilon_j\right)$$

and the calibrated inverse supply function corresponding to $S_i(\mu_i)$ is

$$\mu_i(q) = \bar{\mu}_i \left(1 + \left(q/\bar{S}_i - 1\right)/\eta_i\right)$$

Integrating, we have consumer surplus

$$CS_j(D_j) = \int^{D_j} p_j(q)\mathrm{d}q = \bar{p}_j D_j \left(1 + \left(1 - \frac{D_j}{2\bar{D}_j}\right)/\epsilon_j\right)$$

and total cost

$$TC_i(S_i) = \int^{S_i} \mu_i(q)\mathrm{d}q = \bar{\mu}_i S_i \left(1 + \left(\frac{S_i}{2\bar{S}_i} - 1\right)/\eta_i\right)$$

The Integrated Equilibrium Model

$$\max \sum_j \underbrace{\int^{D_j} p_j(q)\mathrm{d}q}_{CS_j(D_j)} - \sum_i \underbrace{\int^{S_i} \mu_i(q)\mathrm{d}q}_{TC_i(S_i)} - \sum_{ij} c_{ij} X_{ij}$$

s.t.

$$\begin{aligned} S_i &\geq \sum_j X_{ij} && \perp \mu_i \\ \sum_i X_{ij} &\geq D_j && \perp p_j \\ X_{ij} &\geq 0 \end{aligned}$$

Price-Responsive Demand (QCP Formulation)

```
*       Extract the solution with fixed demand as a reference equilibrium:

parameter       muref(i)        Reference marginal cost
                pref(j)         Reference demand price
                sref(i)         Reference supply
                dref(j)         Reference demand
                epsilon(j)      Demand elasticity at the reference point;

muref(i) = supply.m(i); pref(j) = demand.m(j); sref(i) = S.L(i); dref(j) = D.L(j);

epsilon(j) = uniform(0.5, 2);


free variable   SURPLUS         Social surplus;

equation        csurplus        Social surplus with horizontal supply curves (Cs);

csurplus..      SURPLUS =e= -sum((i,j), c(i,j) * X(i,j)) - sum(i, mu(i)*S(i))

        + sum(j,  pref(j) * D(j) * (1 + (1-0.5*D(j)/dref(j)) / epsilon(j)));

model elasticdemand /supply, demand, csurplus/;

*       Remove upper and lower bounds on demand:

D.LO(j) = 0;  D.UP(j) = +inf;

solve elasticdemand using QCP maximizing SURPLUS;
```

QCP Solution

Formulated as a maximization problem, Lagrange multipliers on the supply and demand markets change sign, but they have identical magnitude as compared with the LP solution. This implies that we have "replicated the benchmark equilibrium", having removed upper and lower bounds on demand but introduced the consumer surplus measure which results in no change in prices or quantities.

```
---- EQU supply

        LOWER          LEVEL          UPPER         MARGINAL

1         .              .            +INF           -1.3821
2         .              .            +INF           -1.3298
3         .              .            +INF           -1.2227
4         .              .            +INF           -1.1282
5         .              .            +INF           -1.2468

---- EQU demand

        LOWER          LEVEL          UPPER         MARGINAL

1         .              .            +INF           -1.5539
2         .              .            +INF           -1.2878
3         .              .            +INF           -1.3783
4         .              .            +INF           -1.3969
5         .              .            +INF           -1.3534
```

Price-Responsive Supply and Demand (QCP Formulation)

```
parameter       eta(i)  Price elasticity of supply from node i;

eta(i) = uniform(0.5, 2);


equation        ssurplus        Social surplus with price elastic supply;


ssurplus..      SURPLUS =e= -sum((i,j), c(i,j) * X(i,j))

        + sum(j,  pref(j) * D(j) * (1 + (1-0.5*D(j)/dref(j)) / epsilon(j)))

        - sum(i,  muref(i) * S(i) * (1 + (0.5*S(i)/sref(i)-1)/eta(i)));


model equilibrium /supply, demand, ssurplus/;

*       Remove the upper bound so as to accommodate price-elasticity:

S.UP(i) = +inf;

solve equilibrium using QCP maximizing SURPLUS;
```